

F08HQF (CHBEVD/ZHBEVD) – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

1 Purpose

F08HQF (CHBEVD/ZHBEVD) computes all the eigenvalues, and optionally all the eigenvectors, of a complex Hermitian band matrix. If the eigenvectors are requested, then it uses a divide and conquer algorithm to compute eigenvalues and eigenvectors. However, if only eigenvalues are required, then it uses the Pal–Walker–Kahan variant of the QL or QR algorithm.

2 Specification

```

SUBROUTINE F08HQF(JOB, UPLO, N, KD, AB, LDAB, W, Z, LDZ, WORK,
1              LWORK, RWORK, LRWORK, IWORK, LIWORK, INFO)
ENTRY      chbevd(JOB, UPLO, N, KD, AB, LDAB, W, Z, LDZ, WORK,
1              LWORK, RWORK, LRWORK, IWORK, LIWORK, INFO)
INTEGER    N, KD, LDAB, LDZ, LWORK, LRWORK, IWORK(*),
1          LIWORK, INFO
real      W(*), RWORK(*),
complex  AB(LDAB,*), Z(LDZ,*), WORK(*)
CHARACTER*1 JOB, UPLO

```

The ENTRY statement enables the routine to be called by its LAPACK name.

3 Description

This routine computes all the eigenvalues, and optionally all the eigenvectors, of a complex Hermitian band matrix A . In other words, it can compute the spectral factorization of A as

$$A = Z\Lambda Z^H,$$

where Λ is a real diagonal matrix whose diagonal elements are the eigenvalues λ_i , and Z is the (complex) unitary matrix whose columns are the eigenvectors z_i . Thus

$$Az_i = \lambda_i z_i \quad \text{for } i = 1, 2, \dots, n.$$

4 References

- [1] Golub G H and van Loan C F (1996) *Matrix Computations* Johns Hopkins University Press (3rd Edition), Baltimore

5 Parameters

- 1: JOB — CHARACTER*1 *Input*
On entry: indicates whether eigenvectors are computed as follows:
 if JOB = 'N', then only eigenvalues are computed;
 if JOB = 'V', then eigenvalues and eigenvectors are computed.
Constraint: JOB = 'N' or 'V'.
- 2: UPLO — CHARACTER*1 *Input*
On entry: indicates whether the upper or lower triangular part of A is stored as follows:
 if UPLO = 'U', then the upper triangular part of A is stored;
 if UPLO = 'L', then the lower triangular part of A is stored.
Constraint: UPLO = 'U' or 'L'.

- 3:** N — INTEGER *Input*
On entry: n , the order of the matrix A .
Constraint: $N \geq 0$.
- 4:** KD — INTEGER *Input*
On entry: k , the number of super-diagonals of the matrix A if UPLO = 'U', or the number of sub-diagonals if UPLO = 'L'.
Constraint: $KD \geq 0$.
- 5:** AB(LDAB,*) — **complex** array *Input/Output*
Note: the second dimension of the array AB must be at least $\max(1,N)$.
On entry: the upper or the lower triangle of the n by n Hermitian band matrix A , stored in the first $KD+1$ rows of the array AB. More precisely, the j th column of A is stored in the j th column of the array AB as follows:
 if UPLO = 'U', $AB(KD+1+i-j,j) = a_{ij}$ for $\max(1,j-KD) \leq i \leq j$;
 if UPLO = 'L', $AB(1+i-j,j) = a_{ij}$ for $j \leq i \leq \min(n,j+KD)$.
On exit: A is overwritten by the values generated during the reduction to tridiagonal form. If UPLO = 'U', the first superdiagonal and the diagonal of the tridiagonal matrix are returned in rows KD and $KD + 1$ of the array AB, respectively, and if UPLO = 'L' then the diagonal and the first subdiagonal of the tridiagonal matrix are returned in the first two rows of the array AB.
- 6:** LDAB — INTEGER *Input*
On entry: the leading dimension of the array AB.
Constraint: $LDAB \geq KD+1$.
- 7:** W(*) — **real** array *Output*
Note: the dimension of the array W must be at least $\max(1,N)$.
On exit: the eigenvalues of the matrix A in ascending order.
- 8:** Z(LDZ, *) — **complex** array *Output*
Note: the second dimension of the array Z must be at least $\max(1,N)$ if JOB = 'V', and at least 1 if JOB = 'N'.
On exit: if JOB = 'V', then this is overwritten by the unitary matrix Z which contains the eigenvectors of A . The i th column of Z contains the eigenvector which corresponds to the eigenvalue $W(i)$.
 If JOB = 'N', then Z is not referenced.
- 9:** LDZ — INTEGER *Input*
On entry: the leading dimension of the array Z as declared in the (sub)program from which F08HQF (CHBEVD/ZHBEVD) is called.
Constraints:
 $LDZ \geq \max(1,N)$ if JOB = 'V';
 $LDZ \geq 1$ if JOB = 'N'.
- 10:** WORK(*) — **complex** array *Workspace*
Note: the dimension of the array WORK must be at least LWORK.
On exit: if LWORK > 0, then WORK(1) contains the required minimal size of LWORK.

11: LWORK — INTEGER*Input*

On entry: the dimension of the array WORK as declared in the (sub)program from which F08HQF (CHBEVD/ZHBEVD) is called.

Constraints:

- if $N \leq 1$, then $LWORK \geq 1$;
- if $JOB = 'N'$ and $N > 1$, then $LWORK \geq N$;
- if $JOB = 'V'$ and $N > 1$, then $LWORK \geq 2 \times N^2$.

12: RWORK(*) — *real* array*Workspace*

Note: the dimension of the array RWORK must be at least LRWORK.

On exit: if $LRWORK > 0$, then RWORK(1) contains the required minimal size of LRWORK.

13: LRWORK — INTEGER*Input*

On entry: the dimension of the array RWORK as declared in the (sub)program from which F08HQF (CHBEVD/ZHBEVD) is called.

Constraints:

- if $N \leq 1$, then $LRWORK \geq 1$;
- if $JOB = 'N'$ and $N > 1$, then $LRWORK \geq N$;
- if $JOB = 'V'$ and $N > 1$, then $LRWORK \geq 3 \times N^2 + (4+2k) \times N + 1$ where k is the smallest integer which satisfies $2^k \geq N$.

14: IWORK(*) — INTEGER array*Workspace*

Note: the dimension of the array IWORK must be at least LIWORK.

On exit: if $LIWORK > 0$, then IWORK(1) contains the required minimal size of LIWORK.

15: LIWORK — INTEGER*Input*

On entry: the dimension of the array IWORK as declared in the (sub)program from which F08HQF (CHBEVD/ZHBEVD) is called.

Constraints:

- if $JOB = 'N'$ or $N \leq 1$, then $LIWORK \geq 1$;
- if $JOB = 'V'$ and $N > 1$, then $LIWORK \geq 5 \times N + 2$.

16: INFO — INTEGER*Output*

On exit: $INFO = 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

$INFO < 0$

If $INFO = -i$, the i th parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

$INFO > 0$

If $INFO = i$, then the algorithm failed to converge; i indicates the number of elements of an intermediate tridiagonal form which did not converge to zero.

7 Accuracy

The computed eigenvalues and eigenvectors are exact for a nearby matrix $A + E$, where

$$\|E\|_2 = O(\epsilon)\|A\|_2,$$

and ϵ is the *machine precision*.

8 Further Comments

The real analogue of this routine is F08HCF (SSBEVD/DSBEVD).

9 Example

To compute all the eigenvalues and eigenvectors of the Hermitian band matrix A , where

$$A = \begin{pmatrix} 1.0 + 0.0i & 2.0 + 1.0i & 3.0 + 1.0i & 0.0 + 0.0i & 0.0 + 0.0i \\ 2.0 - 1.0i & 2.0 + 0.0i & 3.0 + 2.0i & 4.0 + 2.0i & 0.0 + 0.0i \\ 3.0 - 1.0i & 3.0 - 2.0i & 3.0 + 0.0i & 4.0 + 3.0i & 5.0 + 3.0i \\ 0.0 + 0.0i & 4.0 - 2.0i & 4.0 - 3.0i & 4.0 + 0.0i & 5.0 + 4.0i \\ 0.0 + 0.0i & 0.0 + 0.0i & 5.0 - 3.0i & 5.0 + 0.0i & 5.0 + 0.0i \end{pmatrix}.$$

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```

*      F08HQF Example Program Text.
*      Mark 19 Release. NAG Copyright 1999.
*      .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX, LDZ, LDAB
      PARAMETER        (NMAX=8,LDZ=NMAX,LDAB=NMAX)
      INTEGER          LWORK, LIWORK, LRWORK
      PARAMETER        (LWORK=NMAX*NMAX+2*NMAX,LIWORK=2+5*NMAX,
+                      LRWORK=4*NMAX*NMAX)
*      .. Local Scalars ..
      INTEGER          I, IFAIL, INFO, J, KD, N
      CHARACTER        JOB, UPLO
*      .. Local Arrays ..
      complex         AB(LDAB,NMAX), WORK(LWORK), Z(LDZ,NMAX)
      real            RWORK(LRWORK), W(NMAX)
      INTEGER          IWORK(LIWORK)
*      .. External Subroutines ..
      EXTERNAL         X04DAF, chbevd
*      .. Intrinsic Functions ..
      INTRINSIC        MAX, MIN
*      .. Executable Statements ..
      WRITE (NOUT,*) 'F08HQF Example Program Results'
*      Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N, KD
      IF (N.LE.NMAX) THEN
*
*      Read A from data file
*
      READ (NIN,*) UPLO
      IF (UPLO.EQ.'U') THEN
        DO 20 I = 1, N
          READ (NIN,*) (AB(KD+1+I-J,J),J=I,MIN(N,I+KD))
20      CONTINUE
      ELSE IF (UPLO.EQ.'L') THEN
        DO 40 I = 1, N
          READ (NIN,*) (AB(1+I-J,J),J=MAX(1,I-KD),I)
40      CONTINUE

```

```

        END IF
*
        READ (NIN,*) JOB
*
*       Calculate all the eigenvalues and eigenvectors of A
*
        CALL chbevd(JOB,UPL0,N,KD,AB,LDAB,W,Z,LDZ,WORK,LWORK,RWORK,
+                LRWORK,IWORK,LIWORK,INFO)
*
        WRITE (NOUT,*)
        IF (INFO.GT.0) THEN
            WRITE (NOUT,*) 'Failure to converge.'
        ELSE
*
*           Print eigenvalues and eigenvectors
*
            WRITE (NOUT,*) 'Eigenvalues'
            DO 60 I = 1, N
                WRITE (NOUT,99999) I, W(I)
60          CONTINUE
            WRITE (NOUT,*)
            IFAIL = 0
*
            CALL X04DAF('General', ' ', N,N,Z,LDZ, 'Eigenvectors', IFAIL)
*
        END IF
        END IF
        STOP
*
99999 FORMAT (3X,I5,5X,2F8.4)
        END

```

9.2 Program Data

F08HQF Example Program Data

```

5 2                                     :Values of N and KD
'L'                                     :Value of UPL0
(1.0, 0.0)
(2.0, 1.0) (2.0, 0.0)
(3.0, 1.0) (3.0, 2.0) (3.0, 0.0)
                (4.0, 2.0) (4.0, 3.0) (4.0, 0.0)
                        (5.0, 3.0) (5.0, 4.0) (5.0, 0.0) :End of matrix A
'V'                                     :Value of JOB

```

9.3 Program Results

F08HQF Example Program Results

Eigenvalues

```

1      -6.4185
2      -1.4094
3       1.4421
4       4.4856
5      16.9002

```

Eigenvectors

	1	2	3	4	5
1	-0.2591	0.6367	0.4516	0.5503	0.1439
	0.0000	0.0000	0.0000	0.0000	0.0000
2	0.0245	-0.2578	-0.3029	0.4785	0.3060
	0.4344	0.2413	-0.4402	0.2759	0.0411
3	0.5159	-0.3039	0.3160	0.2128	0.4681
	-0.1095	-0.3481	0.2978	0.0465	0.2306
4	0.0004	0.3450	-0.4088	-0.1707	0.4098
	-0.5093	-0.0832	-0.3213	0.0200	0.3832
5	-0.4333	-0.2469	0.0204	0.0175	0.1819
	0.1353	0.2634	0.2262	-0.5611	0.5136
